﻿//==============================================================================================
//Author: Uncle Song
//Create Date: 2022-06-28
//Description: 辐射圈
//----------------------------------------------------------------------------------------------
//Alter History:
//              2022-06-28  add Codes.
//============================================================================================== 
Shader "SongShaderDemo/RadiationCircle"
{
    Properties
    {
        _MainTex ("Texture", 2D) = "white" {}
	    _CentorPoint("CentrePoint",Vector)= (0, 0, 0, 0)
		_Dis("Float", Float) = 30//最大距离
		_Range("Range", Range(0, 1.0)) = 0.01//移动范围
		_LineWide("LineWide", Range(0, 1)) = 0.001//线宽
		_Color("color",Color) = (1,1,1,1)   //颜色，一般用fixed4
		_TransColor("TransColor",Color) = (1,1,1,0)   //颜色，一般用fixed4
		_FillTex("FillTexture", 2D) = "white" {}
		_FillColor("FillColor",Color) = (1,1,1,1)   //颜色，一般用fixed4
    }
    SubShader
    {
        Tags { "RenderType" = "Transparent" "IgnoreProjector" = "True" "Queue" = "Transparent" }
        LOD 100
	    Blend SrcAlpha OneMinusSrcAlpha
        Pass
        {
            CGPROGRAM
		
            #pragma vertex vert
            #pragma fragment frag

            #include "UnityCG.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
                float2 uv : TEXCOORD0;
            };

            struct v2f
            {
				float2 uv0 : TEXCOORD0;
                float2 uv : TEXCOORD2;
                float4 vertex : SV_POSITION;
				float3 worldPos : TEXCOORD3;
            };
			CBUFFER_START(UnityPerMaterial)
			sampler2D _MainTex;
			float4 _MainTex_ST;
			sampler2D _FillTex;
			fixed4 _FillColor;
            float4 _FillTex_ST;
			float4 _CentorPoint;
			float _Dis;
			float _Range;
			float _LineWide;
			fixed4 _Color;
			fixed4 _TransColor;
			CBUFFER_END
            v2f vert (appdata v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
                o.uv = TRANSFORM_TEX(v.uv, _FillTex);
				o.uv0 = TRANSFORM_TEX(v.uv, _MainTex);
				o.worldPos = mul(unity_ObjectToWorld, v.vertex).xyz;
				return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {
			     fixed4 col = tex2D(_MainTex, i.uv0);
				float dis = distance(i.worldPos, _CentorPoint.xyz);
			    float rangeMax = _Dis * _Range ;
			    float rangeMin = _Dis*_Range - _LineWide;
				fixed4 Circle= _TransColor;
				fixed4 FillCol = _TransColor;
				if (dis> rangeMin&&dis< rangeMax)
	            {
					Circle= fixed4(_Color.xyz, ((dis - rangeMin) / _LineWide)*(1 - _Range));
	             }
				if (dis < rangeMin)
				{
					fixed4 FillTex = tex2D(_FillTex, i.uv);
					if (FillTex.w!=0)
					{
						FillCol = fixed4(FillTex.xyz, (1 - _Range)) + _FillColor;
					}
					else
					{
						FillCol = FillTex + _FillColor;
					}
					
				}
				fixed4 all = Circle + FillCol;
				return  all*all.w+ col*(1- all.w);
            }
            ENDCG
        }
			
    }
}
